

1. Impure Programming

1. In pure functional OCaml, the only way to have a function with an unconstrained return type is to loop forever, i.e.

```
# let rec f x = f x;;
```

But non-pure OCaml adds another way to do this, what is it?

[small]

2. Doubly-linked lists are lists which have pointers to the previous and next elements, give a type for doubly-linked lists, and the insertion, fold-left and fold-right functions on them. Make both fold-left and fold-right tail-recursive.

Note, you will need to look at == operator in OCaml.

[medium]

3. Discuss how you can detect cycles, and the time and space complexity of this.

[small/medium]

4. It is possible to emulate cyclic lists in OCaml using pure programming too, explain how and contrast it with the impure way.

[small/medium]

5. Why does OCaml not allow one to write `exception Poly of 'a`?

[small]

2. Lazy lists

1. Write a datatype for lazy lists/streams, and a function to create an infinite stream of integers.

[small]

2. Write a function to filter the stream.

[small]

3. You can define the prime number sequence as taking an integer, and removing all multiples of that integer from the rest of the sequence. (Starting at two, of course.) Code the lazy list of primes.

[medium/big]

3. Big question

This question is driven at making a standalone program, for which I'll help write part of it. First, some type definitions for the rest of the question:

```
1 type var_id = int
2 type term =
3   | Var of var_id
4   | List of term list
5   | Int of int
6   | Bool of bool
7   | Str of string;;
```

OCaml

```

8
9  type subst = (var_id * term) list
10 let rec lookup (term: term) (subst: subst) = match term, subst with
11   | Var v, (w, wval)::wvs ->
12     if v=w then wval else lookup (Var v) wvs
13   | _ -> term
14 let extend (varid: var_id) (value: term) (subst: subst) =
15   (varid, value)::subst
16
17 type state = subst * var_id;;
18 type stream = state list;;
19 type goal = state -> stream;;
20
21 (* Some helper functions to abstract away using lists for the stream *)
22 let fail: goal = fun (s, c) -> [];;
23 let success: goal = fun (s, c) -> [(s, c)];;
24 let map = List.map;;
25 let join = String.concat;;

```

1. Some of the past exam questions give you some code, that has been obfuscated, and ask you to explain it, or make modifications to it. Consider the following program, in which the names have been rewritten by a hungry supervisor:

```

1  exception Out_of_ingredients
2  let rec cook pantry pizza pasta =
3    let deep_dish = lookup pizza pantry
4    in let ravioli = lookup pasta pantry
5    in match deep_dish, ravioli with
6    | Var pepperoni, Var cheese when pepperoni=cheese -> pantry
7    | Var pepperoni, _ -> extend pepperoni ravioli pantry
8    | _, Var cheese -> extend cheese deep_dish pantry
9    | List(flour::base), List(tomato::sauce) ->
10      cook (cook pantry flour tomato) (List base) (List sauce)
11    | x, y when x=y -> pantry
12    | _ -> raise Out_of_ingredients;;

```



Describe what it does. You might want to rewrite it using better names. Also, a good place to start is putting type annotations in places.

[small/medium]

2. Using that code, we have the following function:

```

1  let eq a b =
2    fun (s, c) ->
3      try
4        let s' = cook s a b
5        in success (s', c)
6      with Out_of_ingredients -> fail (s, c);;

```



Can you rewrite cook (or your better named version), and eq to not use exceptions?

[small]

3. For the last part of the question, we have some more definitions, and the `var_id` in the state becomes obvious – it represents the first free variable (hence why we are using integers for variable identifiers, rather than say strings).

```
1  let fresh (f: term -> goal) : goal =
2    fun (s, c) -> f (Var c) (s, c + 1);;
3  let mplus (a: stream) (b: stream) : stream = a @ b;;
4  let rec bind (s: stream) (g: goal) : stream =
5    match s with
6    | [] -> []
7    | sub::subs -> mplus (g sub) (bind subs g);;
8
9  let disj (g1: goal) (g2: goal) : goal =
10   fun state -> mplus (g1 state) (g2 state);;
11
12 let conj (g1: goal) (g2: goal) : goal =
13   fun state -> bind (g1 state) g2;;
```

This logic programming language returns a list of all the possible results at once, discuss how you could make it not do so, and how you can then change the search strategies.

To help with this question, I have included some examples of use. For a simple example, we have the code:

```
1  let test =
2    let query =
3      fresh @@ fun p -> fresh @@ fun q ->
4        let l1 = List [p; Int 7]
5        in let l2 = List [Int 9; Int 7]
6        in disj
7          (conj
8            (eq q (Int 5))
9            (eq l1 l2))
10         (eq q (Int 1))
11   in query ([], 0);;
```

which returns

```
1  val test : stream = [([ (0, Int 9); (1, Int 5) ], 2);
2    ([ (1, Int 1) ], 2)];;
```

and this means that there are two solutions to the logic expression

$$(q = 5 \wedge [p, 7] = [9, 7]) \vee q = 1$$

either p is 9 and q is 5, or q is 1 and it doesn't matter what p is.

To give another example, which will help for the search strategy part, consider representing a kind of binary tree where only the leaves hold values, using list terms, for example:

```
1 let t1 = (List [List [List [Int 1; Int 2]; List [Int 3; Int 4]];
2           List [List [Int 5; Int 6]; List [Int 7; Int 8]]);;
```

OCaml

The expression to walk all nodes (leaves and branches) of the tree can be written as follows.

```
3 (* Walk a nested list *)
4 let rec walk term out =
5   disj
6     (fresh @@ fun l -> fresh @@ fun r ->
7       conj
8         (eq term (List[l; r]))
9         (disj (walk l out) (walk r out)))
10    (eq term out);;
```

OCaml

For convenience, I will also define some pretty printing functions, you don't necessarily need to understand them though they are not complicated, just a bit tedious.

```
11 (* Walks substitutions recursively until no more variables can be
12    resolved -- assumes no recursive substitutions *)
13 let rec make_ground subst term : term = match term with
14 | Var v -> (match (List.assoc_opt v subst) with
15   None -> term | Some t -> make_ground subst t)
16 | List l -> List (List.map (make_ground subst) l)
17 | t -> t;;
18
19 (* Formats a term *)
20 let rec string_of_term = function
21 | Var v -> Printf.sprintf "V%d" v
22 | Int i -> Printf.sprintf "%d" i
23 | Str s -> String.escaped s
24 | Bool b -> if b then "true" else "false"
25 | List l -> "[" ^ String.concat "; " (List.map string_of_term l) ^ "];";
26
27 let print_all_mappings v streams =
28   streams
29   |> map (fun (subst, _) -> string_of_term @@ make_ground subst v)
30   |> join "\n"
31   |> print_endline;;
```

OCaml

Finally we can walk the tree and print the variable binding to the second argument of walk.

```
1 walk t1 (Var 0) ([], 1)
2 |> print_all_mappings (Var 0);;
```

OCaml

```
1
2
[1; 2]
3
4
[3; 4]
```

```

[[1; 2]; [3; 4]]
5
6
[5; 6]
7
8
[7; 8]
[[5; 6]; [7; 8]]
[[[1; 2]; [3; 4]]; [[5; 6]; [7; 8]]]

```

[medium]

3.1. Bonus

This uses code from the “Big question” section, optionally along with the lazy list version, to make a type checker for the simply typed lambda calculus. First, we will need some convenience functions.

```

1 let any (g: goal list) : goal =
2   List.fold_left disj fail g;;
3
4 let all (g: goal list) : goal =
5   List.fold_left conj success g;;

```


OCaml

The simply typed lambda-calculus has the typing rules

$$\begin{array}{c}
 \frac{\Gamma, \text{arg} : \alpha \vdash \text{body} : \beta}{\Gamma \vdash \text{fn arg} \rightarrow \text{body} : \alpha \rightarrow \beta} \text{ (fun)} \\
 \frac{\Gamma \vdash \text{func} : \alpha \rightarrow \beta \quad \Gamma \vdash \text{arg} : \alpha}{\Gamma \vdash \text{func arg} : \beta} \text{ (app)} \\
 \frac{}{\Gamma, x : \alpha \vdash x : \alpha} \text{ (var. found)} \\
 x \neq y \frac{\Gamma \vdash x : \beta}{\Gamma, y : \alpha \vdash x : \beta} \text{ (var. not found)}
 \end{array}$$

Which we can easily express as a series of logic rules in our simple logic programming system.

```
6 let rec typecheck gamma term typ =
7   any [
8     (* fun *)
9     (fresh @@ fun var -> fresh @@ fun body ->
10      fresh @@ fun a -> fresh @@ fun b ->
11       all [
12         eq term @@ List[Str "fun"; var; body];
13         eq typ @@ List[a; Str "->"; b];
14         typecheck ((var, a)::gamma) body b]);
15     (* app *)
16     (fresh @@ fun fn -> fresh @@ fun arg ->
17      fresh @@ fun a -> fresh @@ fun b ->
18       all [
19         eq term @@ List[fn; arg];
20         eq typ b;
21         typecheck gamma fn (List[a; Str "->"; b]);
22         typecheck gamma arg a]);
23     match gamma with
24     | [] -> fail
25     | (x, a)::rest ->
26       any [
27         (* var found *)
28         all [eq x term; eq a typ];
29         (* var not found *)
30         typecheck rest term typ]
31   ];;
```



We will also write a couple of functions to pretty-print the resulting type, to make it easier to read.

```

32 (* Creates a module for mapping variables (integers) *)
33 module VarMap = Map.Make(struct include Int end);;
34
35 (* Maps a number to a type, e.g. 0 to 'a, 25 to 'z, 26 to 'aa, 27 to 'ab etc *)
36 let format_var n =
37   let char_of_var n = Char.chr (Char.code 'a' + n)
38   in let rec inner n acc =
39     if n < 26
40     then char_of_var n :: acc
41     else inner (n / 26 - 1) @@ char_of_var (n mod 26) :: acc
42   in "" ^ (String.of_seq @@ List.to_seq @@ inner n []);;
43
44 (* Map each variable to 'a, 'b, etc *)
45 let map_var vars v = match VarMap.find_opt v vars with
46 | None ->
47   let mapping = format_var (VarMap.cardinal vars)
48   in (VarMap.add v mapping vars, mapping)
49 | Some s -> (vars, s);;
50
51 (* Formats a type (which is assumed to be only `List[type; Str"->" type]` or
52   `Var v`) as a readable string *)
53 let rec format_type vars term = match term with
54 | List[List l; Str "->"; r] ->
55   let (vars', l') = format_type vars (List l)
56   in let (vars'', r') = format_type vars' r
57   in (vars'', Printf.sprintf "(%s) -> %s" l' r')
58 | List[l; Str "->"; r] ->
59   let (vars', l') = format_type vars l
60   in let (vars'', r') = format_type vars' r
61   in (vars'', Printf.sprintf "%s -> %s" l' r')
62 | Str s -> (vars, s)
63 | Var v -> map_var vars v
64 | _ -> failwith "format_type";;
65
66 let format_results query results =
67   map (fun ((subst, _): state) ->
68     let (_, ty) = format_type VarMap.empty (make_ground subst query)
69     in ty) results
70   |> join "\n" |> Printf.printf "Results:\n%s\n";;

```

Now we can try some examples! This fails because it has an unbound variable

```

let t0 = typecheck [] (Str"x") (Var 0) ([], 1);;
format_results (Var 0) t0;;

```

Results:

If we bind the variable, it works:

```
let t1 = typecheck [(Str"x", (Var 0))] (Str"x") (Var 0) ([], 1);;
format_results (Var 0) t1;;
Results:
'a
```

For a different expression that has no unbound variables:

```
let t2 = typecheck [] (List[Str"fun"; Str"x"; Str"x"]) (Var 0) ([], 1);;
format_results (Var 0) t2;;
Results:
'a -> 'a
```

And for a more complicated expression:

```
let t3 = typecheck
  []
  (* Encoding the term `fun x -> fun y -> fun z -> x y (z y)` *)
  (List[Str"fun"; Str"x";
    List[Str"fun"; Str"y";
      List[Str"fun"; Str"z";
        List[
          List[Str"x"; Str"y"];
          List[Str"z"; Str"y"]]]]]])
  (Var 0)
  ([], 1);;
format_results (Var 0) t3;;
Results:
('a -> 'b -> 'c) -> 'a -> ('a -> 'b) -> 'c
('a -> 'b -> 'c) -> 'a -> ('a -> 'b) -> 'c
```

Note, because we do not do occurs checks, the following also works (not pretty-printing it because the pretty-printer would try to expand it forever):

```
let t4 = typecheck
  []
  (* Encoding the term `fun x -> x x` *)
  (List[Str"fun"; Str"x"; List[Str"x"; Str"x"]])
  (Var 0)
  ([], 1);;
val t4 : stream =
  Cons
    (([(7, List [Var 7; Str "->"; Var 8]); (3, List [Var 7; Str "->"; Var 8]);
      (4, Var 8); (6, Str "x"); (5, Str "x");
      (0, List [Var 3; Str "->"; Var 4]); (2, List [Str "x"; Str "x"]);
      (1, Str "x")],
      9),
    <fun>)
```

Which does represent the infinite type `'a = 'a -> 'b` as expected. To fix this, you would have to look at adding an “occurs check” into the extend function.

If you want to learn more about this kind of logic programming, you could read more about “microkanren”, or read the book “Reasoned Schemer” – it is an excellent book, I recommend it. So are the rest of that series.